Current distribution in a spiral FFAG

Bruno Autin

bautin1@club-internet.fr

Introduction

The design of RACCAM has to opt between a combined function magnet and a magnet where the field law is determined by pole face windings. This paper addresses the second option.

Field produced by a conductor in the median plane



Fig.1 Geometry of a thin conductor and of the observation orbit

The geometry is cylindrical (Fig.1). Both conductor and observation orbit are arcs of circle. The conductor is infinitely thin and located against the pole at a height g. The orbit lies in the median plane. For a magnet of opening ω, an orbit of radius r has the length $\omega\, r$. The origin and extremity of the arc have the azimuths

$$\theta_1 = Log\left(\frac{r}{a}\right) tan\,(\xi)$$
$$\theta_2 = \omega + Log\left(\frac{r}{a}\right) tan\,(\xi)$$

for a magnet of opening angle ω, minimum radius a and spiral angle ξ. A current I passes through the element of length dl located about the point M conductor. It creates at the point A a field dB given by Biot and Savart's law:

$dB = \frac{\mu_0 I}{4\pi} \frac{dl \wedge u}{d^2}$

where u is the unit vector of the vector $\vec{MA}$ and d the distance from M to A. The coordinates of M are

$$\begin{pmatrix} R\cos\theta \\ R\sin\theta \\ g \end{pmatrix},$$

those of A are

$$\begin{pmatrix} r\cos\alpha \\ r\sin\alpha \\ 0 \end{pmatrix}$$

so that the vector u is

$$\frac{1}{d}\begin{pmatrix} r\cos\alpha - R\cos\theta \\ r\sin\alpha - R\sin\theta \\ -g \end{pmatrix}$$

with

$$d = \sqrt{r^2 + R^2 + g^2 - 2\, r\, R\cos(\alpha - \theta)}$$

The vector dl is

$$\begin{pmatrix} -\sin\theta \\ \cos\theta \\ 0 \end{pmatrix} R\, \mathrm{d}\theta.$$

The magnetic field is thus:

$$\mathrm{d}B = \frac{\mu_0 I}{4\pi d^3}\begin{pmatrix} -\, g\cos\theta \\ -\, g\sin\theta \\ R -\, r\cos(\alpha -\, \theta) \end{pmatrix} R\, \mathrm{d}\theta.$$

The field produced by the conductor symmetric with respect to the median plane is obtained by changing g into -g. The total field is thus purely vertical and, if one adds the effect of the images in the poles, the full vertical field is

$$B_z = \frac{\mu_0 I R}{\pi} \int_{\theta_1}^{\theta_2} \frac{R - r\cos(\alpha - \theta)}{(r^2 + R^2 + g^2 - 2\, r\, R\cos(\alpha - \theta))^{\frac{3}{2}}}\, d\theta.$$

It turns out, using *Mathematica*, that the integral can be expressed as a closed expression in terms of elliptic functions

$$B_z = \frac{\mu_0 I}{\pi r}(F(\theta_2) - F(\theta_1))$$

with

$$F(\theta) = \frac{1}{\sqrt{\frac{g^2}{r^2}+\left(1-\frac{R}{r}\right)^2}\left(\frac{g^2}{r^2}+\left(1+\frac{R}{r}\right)^2\right)}\left(\left(1+\frac{g^2-R^2}{r^2}\right)\text{EllipticE}\left[\frac{\alpha-\theta}{2},\,-\frac{4R}{r\left(\frac{g^2}{r^2}+\left(1-\frac{R}{r}\right)^2\right)}\right]-\right.$$

$$\left.\left(\frac{g^2}{r^2}+\left(1+\frac{R}{r}\right)^2\right)\text{EllipticF}\left[\frac{\alpha-\theta}{2},\,-\frac{4R}{r\left(\frac{g^2}{r^2}+\left(1-\frac{R}{r}\right)^2\right)}\right]+\frac{2R\left(1+\frac{g^2-R^2}{r^2}\right)\text{Sin}[\alpha-\theta]}{r\sqrt{\left(\frac{g^2}{r^2}+\left(1-\frac{R}{r}\right)^2\right)\left(1+\frac{g^2+R^2}{r^2}-\frac{2R\,\text{Cos}[\alpha-\theta]}{r}\right)}}\right).$$

The shape of that field is shown in Fig.2 for the dependence on the radius and on Fig.3 for the dependence on the azimuth. The field is, as expected, *0* below the conductor and has a maximum on the orbit at a distance equal to half a gap from the conductor orbit. The orbit of the conductor is located at a raius of 2 m.



Fig.2   Radial variation of the field

The azimuthal variations are shown for minimum and maximum radii (left hand column) and for the radii of extremum field (right hand column). The blue dots indicate the spiral borders of the magnet and provide evidence for a strong upstream-downstream assymmetry of the field variations. In the core of the magnet, the field is fairly constant. An overshoot outside the magnet is also noticeable for large radii.



Fig.3 Azimuthal variations of the field

Field produced by a loop of conductors in the median plane

The usual method to calculate a current distribution consists of building a matrix A whose each coefficient is the field produced by a unit current at a given observation point and solving the linear equation

$A.\mathbf{I} = \mathbf{B}$

where **I** is the unknown current vector and **B** the prescribed field values at the observation points. The big problem is to get a well conditioned matrix A. The field law of a single conductor produces actually an ill-conditioned matrix. A technique to avoid that computational difficulty lies in grouping the conductors in loops and evaluating the individual currents by applying the superposition principle. The loop is made of two conductors: one of current I and the second of current $-I$ shifted in the radial direction by the distance Δr. In Fig.4, the first conductor occupies the same position as the conductor which produces the blue field ($r = 2$ in that special example) and the second conductor is shifted outwards by the half gap g (3 cm here). The width of a loop and the distance between loops certainly require detailed investigation but here, it is assumed that the width is g and the distance $g/2$. This way two adjacent loops fully overlap and the third loop has a commom conductor with the first one but with an opposite current.



Fig.4 Comparison between the field produced by a single conductor (blue) and a loop (red).

Current distribution

The actual determination of the current distribution relies on a number of *Mathematica-Geometrica* functions. The loop method is applied to loops of width g/n with one common conductor; in other words they are spaced by g/n. For this reason there are two more individual currents than loop currents. In order to improve the field quality in the aperture, the loops extend from $R_{min}-4g$ to $R_{max}+2g$. The observation points are chosen where the field due to a loop has its maximum effect. Their radius is determined numerically. As the matrix A is well conditioned, there are as many observation points as loops and they are located on the mean azimuth of each arc.

■ *Mathematica* - *Geometrica* functions

`FFAGSpiralParameters` parameters list of a proton spiral FFAG made of the following replacement rules
NumberOfCells → *n* number of cells,
PackingFactor → *p* ratio of the opening of a magnet to the opening of a cell,
InjectionKineticEnergy → E_1 kinetic energy at injection in MeV,
EjectionKineticEnergy → E_2, kinetic energy at ejection in MeV,
EjectionRadius → R_2 radius of ejection orbit in meter,
FieldPower → *k* power of the field law,
MagnetOpening → ω opening of the magnet,
SpiralAngle → ξ spiral angle,
MagnetHalfGap → *g* magnet half gap,
InjectionField → B_1 field on injection orbit in Tesla,
InjectionRadius → R_1 radius of injection orbit in meter,
EjectionField → B_2 field on ejection orbit in Tesla,

`FFAGSpiral`[*options*] layout and new parameters list of a proton spiral FFAG with the following options at input.,
MachineCenter → *c* enter of the machine. Default: `CPoint`[0,0]
InitialAngle → ϕ angle of the radius joining the center of the machine to the lower right corner of the first sector.
Segment → *v* If *v* is `True`, each sector is returned as a curvilinear polygon and can thus be painted. If *v* is `False` (default), each sector is returned as a list of four curves: the two arcs and the two spirals.
At output, a graph of the machine is produced and a new parameters list is generated.
FFAGSpiralGraph → *graph*, see the option *Segment*,
`FFAGSpiralParameters` in `FFAGSpiralParameters` the last three parameters are not independent, they are infered from the first nine parameters of the list.

`FFAGSpiralArc`[*n*, *R*, *options*] *n* arcs at radius *R* inside the sectors of a spiral FFAG.

`PolarConductorField`[r, α, R, θ_1, θ_2] field created at point of polar coordinates (r, α) by an arc of conductor of radius *R* located in a sector of half gap *g* and limited by angles θ_1 and θ_2.
The center of the machine is `CPoint`[0,0]. The initial angle is 0. The half gap is taken from `FFAGSpiralParameters`.

`LoopField`[r, α, R, δ] field created at point of polar coordinates (r, α) by a loop made of an arc of conductor of radius *R* and of a second arc traversed by an opposite current and of radius $R + \delta$.
The center of the machine is `CPoint`[0,0]. The initial angle is 0. The other parameters are taken from `FFAGSpiralParameters`.

`CurrentDistribution`[*n*, *option*] current distribution in the sector of a spiral FFAG using the loop technique. Each loop has a width *g* and the distance between adjacent loops is *g*/*n*, *n* being an integer. The option is
FieldCorrection → *f*, *f* is a pure function applied to the reference field law *B*(*r*) in the form *B*(*r*)(1-*f*(*r*)). The default value of *f*: (0&) corresponds to an absence of correction.
f is obtained as a fit to the residual field errors in the aperture when the reference field is entered without correction.
DisplayFunction → *d*, displays the plots of the output if *d* is `$DisplayFunction` (default) and inhibits the display if *d* is `Identity`.
The output is a list of substitution rules:
LoopCurrents → *l*, list of loop currents.

Currents → *i*, list of individual currents.
FieldCurrentMatrix → *m*, matrix of the fields in mGauss produced by unit currents in each conductor.
FieldErrorFunction → *fe*, pure function of the relative field error $\frac{B(r)-B_0(r)}{B_0(r)}$ where $B(r)$ is the field produced by the currents and $B_0(r)$the reference field law.
The azimuth where the field is observed is at the midpoint of the arc of radius r.
Field3DFunction → *f3D*, pure function of the field produced by arcs of conductors as a function of radius r and azimuth α.
Layout → λ, layout of the conductors and observation points.
LoopCurrentPlot → *lp*, plot of the loop currents at each first conductor radius.
CurrentPlot → *ip*, plot of the currents at each conductor radius.
FieldPlot → *fp*, plot of the reference and actual fields.
FieldErrorPlot → *fep*, plot of the field errors.
FieldPlot3D → *fp3D*, plot of the actual field produced by arcs of conductors as a function of radius r and azimuth α.
The field is plotted in a region of extension $3\omega/2$ upstream and downstream of the magnet so that, including the magnet itself, the total angle region is $5\omega/2$..

■ Parameter list

The parameters are those of the RACCAM project studied at LPSC:
$E_1 = 17$ Mev,
$E_2 = 180$ Mev,
$n = 8$,
$p = 0.38$,
$R_2 = 3.4825$ m,
$\omega = 17.1\ \pi/180$,
$\xi = 49.825\ \pi/180$,
$g = 0.03$ m,
$k = 4.385$.
The function `FFAGSpiral` completes the above list with the three extra parameters and provides the layout of the machine:
$R_1 = 2.7759$ m
$B_1 = 0.55455$ T
$B_2 = 1.499$ T



Fig.5 Layout of a spiral FFAG (RACCAM).

■ Current distribution for a loop width equal to *g*

The layout of the conductors and observation points, the loop and individual current distributions are shown in Figs. 6 to 8. The field and relative field errors in Figs. 9 and 10. It must be noticed that the field functions are continuous and calculated analytically in `CurentDistribution`. The residual errors exceed 1% and may not be acceptable for a good particle optics. A new test will be performed with loops of smaller width.



Fig.6 Layout of observation points (blue) and conductors (red).



Fig.7 Loop current distribution.



Fig.8 Individual current distribution. The blue dots represent the limits of aperture.



Fig.9 Reference field (blue) and actual field (magenta) produced by individual currents.



Fig.10 Relative field error.

■ Current distribution for a loop width equal to *g*/2

The same calculations as above are resumed with a loop width equal to $g/2$. The number of loops and observation points are doubled with respect to the previous case (Fig. 11). The loop currents oscillate (Fig. 12) but the individual currents (Fig. 13), yet stronger, have the same distribution as before. The objective of reducing the relative field error is partially reached (Fig. 14) since the field error (Fig. 15) has no longer rapid variations and is prone to further correction.



Fig.11 Layout of observation points (blue) and conductors (red).



Fig.12 Loop current distribution.



Fig.13 Individual current distribution. The blue dots represent the limits of aperture.



Fig.14 Reference field (blue) and actual field (magenta) produced by individual currents.



Fig.15 Relative field error.

■ Anticipating residual field errors

The field errors can be precisely described by fitting a polynomial. In the present case, a good fit is obtained with the function

$f = 47.5702 - 59.9461\, r + 28.1993\, r^2 - 5.86999\, r^3 + 0.456308\, r^4$

The principle of correction consists then of anticipating the errors due to the currents by modifying the reference field according to the expression

$B_1\left(\frac{r}{R_1}\right)^k$(1-f).

The results of the new computation are shown in Figs. 15 and 16. The field errors are almost within one per mil in absolute value. It is possible to make a new correction to the field law to still improve the field fluctuations but that refinement does not seem to be necessary.



Fig.16 Relative field error after alteration of the field law.



Fig.17 Same function as in Fig. 16 but observed in the beam aperture region with a zoom of 100.

The list of the 63 individual currents is displayed in Table 1. In the non oscillating region, the current varies monotonically from -11 to 32 kA. Such intensities are certainly achievable with conventional technology. Of more concern are the large oscillations at each edge of the aperture. Those oscilations are physical and not due to some numerical artifact. They may be reduced and even eliminated by shaping the edges of the magnet. This is a common technique ("shimming") traditionally based on conformal mapping for the theoretical part and 2D computations for the numerical part. It is however beyond the scope of that paper.

```
{-789075., 555306., -363177., 105426., -110110., -7515.56, -51833.2, -29146.4,
 -37500.4, -31849.2, -32782.6, -30805.6, -30227.9, -29038.6, -28134., -27085.7,
 -26080.6, -25029.2, -23972.5, -22893.5, -21802.5, -20696.8, -19579.2,
 -18449., -17306.6, -16151.1, -14981.7, -13797.4, -12597., -11379., -10142.3,
 -8885.44, -7607.28, -6306.76, -4983.05, -3635.58, -2264.17, -869.026, 549.244,
 1989.61, 3450.92, 4931.91, 6432.45, 7952.68, 9497.53, 11069.8, 12688.6,
 14351.1, 16118.1, 17926.4, 20002.3, 21950.3, 24763.7, 26395.5, 31533.6,
 30054.2, 44556.2, 25527.7, 82861.5, -28267.4, 260540., -491080., 1.09116×10^6}
```

Table1 List of the individual currents.

3D Field

The field created in the whole magnetic volume is shown in Fig.18. Although the calculation is rigorous, this field is only due to the conductors located inside the magnet, some model of return conductors at the edge should be added to get a more complete picture. A particle can be tracked analytically in the horizontal median plane using the pure function `Field3DFunction` returned by `CurrentDistribution`. It is likely that the theoretical spiral edges will have to be re-designed. Then the method used in the previous section for reducing the field error variations can be applied to produce a perturbed spiral edge which will restore the wanted optical properties.



Fig.18 Magnetic field B as a function of radius r and azimuth α.

Conclusion

A stable procedure of calculation of current distribution has been elaborated for a spiral magnet of FFAG accelerator. A technique to reduce the residual field fluctuations along the midline of the magnet within the per mil range has been elaborated. The analytical function which describes the vertical field as a function of radius and azimuth can be used for analytical tracking in the median plane opening the way to the design of new edge profiles, slightly different from the original spirals, to obtain wanted optical properties. Missing is the geometry of the return conductors and an extension of the theoretical treatment to the horizontal field so that the vertical motion could also be determined. Last and not least, though the central currents have a regular distribution and can certainly be achieved, the end currents have excessive values and a proper design of the pole shape in that regions is mandatory to get a complete technological design of the magnet.

Code of the functions

```
In[1]:= << BeamOptics`BeamOptics06`

In[1038]:=
Clear[FFAGSpiral, FFAGSpiralArc, FFAGSpiralArcAngle, CurrentDistribution,
 NewCurrentDistribution, PolarConductorField, LoopField]
Attributes@PolarConductorField = {Listable}
Attributes@LoopField = {Listable}
FFAGSpiralParameters = {InjectionKineticEnergy → 17, InjectionField → if,
  InjectionRadius → ir, EjectionKineticEnergy → 180, NumberOfCells → 8,
  PackingFactor → .38, FieldPower → 4.385, EjectionField → ef, EjectionRadius → 3.4825,
  MagnetOpening → 17.1 π / 180, MagnetHalfGap → .03, SpiralAngle → 49.825 π / 180}
Options[FFAGSpiral] = {InitialAngle → 0, MachineCenter → CPoint[0, 0], Segment → False}

FFAGSpiral[opts___Rule] :=
 Module[{a1, a2, ar, ar1, ar2, ce, e1, e2, ef, f, index, g, gr, if,
   ir, k, l, m1, m2, mom1, mom2, n, p1, p2, pf, q1, q2, r1, r2, rho1,
   rho2, s, s1, s2, sp1, sp2, tt, t1, t2, uu, vv, ww, x, y, ω, ξ},
  {ce, t1, tst} = {MachineCenter, InitialAngle, Segment} /. {opts} /.
    Options[FFAGSpiral];
  {n, pf, e1, e2, r2, index, ω, ξ, g} = {NumberOfCells, PackingFactor,
     InjectionKineticEnergy, EjectionKineticEnergy, EjectionRadius, FieldPower,
     MagnetOpening, SpiralAngle, MagnetHalfGap} /. FFAGSpiralParameters;
  {if, ir, ef} = {InjectionField, InjectionRadius, EjectionField} /.
    FFAGSpiralParameters;
  {x, y} = List @@ ce;
  q1 = Table[t1 + k 2 π / n, {k, 0, n - 1}];
  {mom1, mom2} = Momentum /. (LParameters[Proton, KineticEnergy[#]] & /@ {e1, e2});
  rho2 = r2 Sin[pf π / n] / Sin[ π / n];
  rho1 = rho2 (mom1 / mom2) ^ (1 / (index + 1));
  r1 = rho1 Sin[ π / n] / Sin[pf π / n];
  sp1 = r1 Exp[(# - q1)/Tan[ξ]];
  q2 = q1 + ω;
  sp2 = r1 Exp[(# - q2)/Tan[ξ]];
  t1 = q1 + Tan[ξ] Log[r2 / r1];
  t2 = q2 + Tan[ξ] Log[r2 / r1];
  pr1 = Thread[f[Transpose[{q1, t1}]]] /. f[l_] → Rule[PRange, Join[l, {25}]];
  m1 = Map[Function, PPoint[x + sp1 Cos[#], y + sp1 Sin[#]], {2}];
  p1 = MapThread[Append, {m1, pr1}];
  pr2 = Thread[f[Transpose[{q2, t2}]]] /. f[l_] → Rule[PRange, Join[l, {25}]];
  m2 = Map[Function, PPoint[x + sp2 Cos[#], y + sp2 Sin[#]], {2}];
  p2 = MapThread[Append, {m2, pr2}];
  a1 = Arc[ce, Pointer[p1, q1], ω];
  ar1 = Segment /@ a1;
  a2 = Arc[ce, Pointer[p1, t1], ω];
  ar2 = Reverse /@ (Segment /@ a2);
  ar = Transpose[{ar1, ar2}];
  s1 = Reverse /@ (Segment /@ p1);
  s2 = Segment /@ p2;
  s = Transpose[{s1, s2}];
  gr =
   If[tst, Thread[List[s, ar]] /. ({{uu_, vv_}, {ww_, tt_}} → Join[uu, ww, vv, tt]),
        Transpose[{a1, a2, p1, p2}]];
  {if, ef} = {mom1 / rho1, mom2 / rho2} / 300;
  FFAGSpiralParameters =.;
```

```
  param = {NumberOfCells → n, PackingFactor → pf,
    InjectionKineticEnergy → e1, EjectionKineticEnergy → e2, EjectionRadius → r2,
    FieldPower → index, MagnetOpening → ω, SpiralAngle → ξ, MagnetHalfGap → g,
    InjectionField → if, InjectionRadius → r1, EjectionField → ef};
  {FFAGSpiralGraph → gr, FFAGSpiralParameters = param}]

FFAGSpiralArc[n_, r2_, opts___] :=
 Module[{ar, ar1, ar2, ce, f, k, l, m1, m2, p1, p2, q1, q2,
   r1, s, s1, s2, sp1, sp2, tt, t1, t2, uu, vv, ww, x, y, ω, ξ},
  {ce, t1} = {MachineCenter, InitialAngle} /. {opts} /. Options[FFAGSpiral];
  {r1, ω, ξ} = {InjectionRadius, MagnetOpening, SpiralAngle} /. FFAGSpiralParameters;
  {x, y} = List @@ ce;
  q1 = Table[t1 + k 2 π / n, {k, 0, n - 1}];
  sp1 = r1 Exp[(# - q1)/Tan[ξ]];
  t1 = q1 + Tan[ξ] Log[r2 / r1];
  pr1 = Thread[f[Transpose[{q1, t1}]]] /. f[l_] → Rule[PRange, Join[l, {25}]];
  m1 = Map[Function, PPoint[x + sp1 Cos[#], y + sp1 Sin[#]], {2}];
  p1 = MapThread[Append, {m1, pr1}];
  PPoint[r2 Cos[#1] &, r2 Sin[#1] &, PRange → {Log[r2/r1] Tan[ξ], ω + Log[r2/r1] Tan[ξ] }]]

FFAGSpiralArcAngle[n_, r2_, opts___] := Module[{r1, ω, ξ},
  {r1, ω, ξ} = {InjectionRadius, MagnetOpening, SpiralAngle} /. FFAGSpiralParameters;
  ω/2 + Log[r2/r1] Tan[ξ] ]

CurrentDistribution[n_, opts___Rule] :=
 Module[{a, an, anc1, anc2, ani1, ani2, arcs, arcsc, b, b1, bcor,
   bn, df, field, field3d, fielderr, g, iloop, in, index, l0, l1, l2, l3,
   loopfield, m, mat, nr, pol, r, rc, ri, ro, txi, u, v, x, y, α, ω, ξ},
  {bcor, df} = {FieldCorrection, DisplayFunction} /. {opts} /.
    {FieldCorrection → (0 &), DisplayFunction → $DisplayFunction};
  {a, b, ω, ξ, g, index, b1} = {InjectionRadius, EjectionRadius,
     MagnetOpening, SpiralAngle, MagnetHalfGap,
     FieldPower, InjectionField} /. FFAGSpiralParameters;
  txi = Tan[ξ];
  (*1 boucle de largeur g/n tous les g/n. Pour une meilleure qualité du chamo,
   les conducteurs sont écartés de 4 et 2 g par rapport à l' ouverture.*)
  {a, b} = {a - 4 g, b + 2 g};
  nr = Ceiling[n (b - a) / g];
  {arcs, rc, an} = Transpose@Table[{FFAGSpiralArc[1, a + m * g / n],
      a + m * g / n, FFAGSpiralArcAngle[1, a + m * g / n]}, {m, 0, nr}];
  {anc1, anc2} = Transpose[(PRange /. #) & /@ (List @@@ (Drop[#, 2] & /@ arcs))];
  lf = LoopField[r, an, rc, g / n];
  (*point d' observation placé au champ maximum
    produit par la boucle. Pour que FindRoot trouve le vrai maximum
    qui est légèrement supérieur à r, g/10 est ajouté à rc.*)
  ro = Chop[r /. MapThread[FindRoot[D[#1, r] == 0, {r, #2}] &, {lf, rc + g / 10}]];
  mat = Table[LoopField[ro, an, rc[[m]], g], {m, nr + 1}];
  bn =  (b1 10^7 (ro / a) ^index) (1 - bcor /@ ro);
  iloop = LinearSolve[Transpose[mat], bn];
  l0 = Table[0, {n}];
  in = Join[iloop, l0] + Join[l0, -iloop];
  l1 = Table[Last[rc] + m g / n, {m, 1, n}];
  l2 = Table[FFAGSpiralArcAngle[1, a + m * g / n], {m, nr + 1, nr + n}] - ω / 2;
  l3 = l2 + ω;
  ri = Join[rc, l1];
  ani1 = Join[an - ω / 2, l2];
  ani2 = Join[an + ω / 2, l3];
```

```
  arcsc = Flatten@Table[FFAGSpiralArc[1, a + m * g / n], {m, nr + 1, nr + n}];
  loopfield = 10^-7 LoopField[r, FFAGSpiralArcAngle[1, r], rc, g].iloop;
  field =
   10^-7 PolarConductorField[r, FFAGSpiralArcAngle[1, r], ri, ani1, ani2, g].in;
  field3d = (10^-7 PolarConductorField[r, α, ri, ani1, ani2, g].in)
    Boole[Log[r / a] txi - 3 ω / 2 ≤ α ≤ Log[r / a] txi + 5 ω / 2 ];
  fielderr = Evaluate[(field / (b1 (r / a)^index) - 1) /. r → #] &;
  field3df = Evaluate[Chop[field3d /. r → #1 /. α → #2]] &;
  {LoopCurrents → iloop,
   Currents → in,
   FieldCurrentMatrix → mat,
   FieldErrorFunction → fielderr,
   Field3DFunction → field3df,
   Layout → Draw[Blue,
     CPoint[ro Cos[an], ro Sin[an]], Red, arcs, arcsc, DisplayFunction → df],
   LoopCurrentPlot → ListPlot[Thread[List[rc, iloop 10^-3]],
     PlotRange → All, PlotJoined → True, Axes → True,
     AxesLabel → {"r  [m]", "I  [kA]"}, PlotStyle → Blue, DisplayFunction → df,
     Epilog → {Blue, PointSize[.02], Point[{a + 4 g, 0}], Point[{b - 2 g, 0}]}],
   CurrentPlot → ListPlot[Thread[List[ri, in 10^-3]], PlotRange → All,
     PlotJoined → True, Axes → True, AxesLabel → {"r  [m]", "I  [kA]"},
     PlotStyle → Red, DisplayFunction → df,
     Epilog → {Blue, PointSize[.02], Point[{a + 4 g, 0}], Point[{b - 2 g, 0}]}],
   FieldPlot → Plot[{b1 (r / a)^index, loopfield, field}, {r, a, Last[ri]},
     PlotRange → All, AxesLabel → {"r  [m]", "B  [T]"},
     PlotStyle → {Blue, Red, Magenta}, DisplayFunction → df,
     Epilog → {Blue, PointSize[.02], Point[{a + 4 g, 0}], Point[{b - 2 g, 0}]}];
   FieldPlot3D → Plot3D[field3d, {r, a, b}, {α, -3 ω / 2, Log[b / a] txi + 5 ω / 2},
     PlotRange → All, AxesLabel → {"r  [m]", "α  [rad]", "B  [T]"},
     PlotPoints → 100, DisplayFunction → df];
   FieldErrorPlot → Plot[field / (b1 (r / a)^index) - 1, {r, a, Last[ri]},
     PlotRange → {-.1, .1}, AxesLabel → {"r  [m]", "ΔB/B"},
     PlotStyle → Magenta, DisplayFunction → df,
     Epilog → {Blue, PointSize[.02], Point[{a + 4 g, 0}], Point[{b - 2 g, 0}]}]}
 ]
```

```
PolarConductorField[r_, α_, R_, θ1_, θ2_, g_] := Module[{θ, f, γ = g / r, ρ = R / r},
```

$$f = \frac{4}{r\sqrt{\gamma^2+(1-\rho)^2}\ (\gamma^2+(1+\rho)^2)}\left((1+\gamma^2-\rho^2)\ \mathrm{EllipticE}\left[\frac{\alpha-\theta}{2}, -\frac{4\rho}{\gamma^2+(1-\rho)^2}\right] - \left(\gamma^2+(1+\rho^2)^2\right)\mathrm{EllipticF}\left[\frac{\alpha-\theta}{2}, -\frac{4\rho^2}{(\gamma^2+(1-\rho^2)^2)}\right] + \frac{2\rho\ (1+\gamma^2-\rho^2)\ \mathrm{Sin}[\alpha-\theta]}{\sqrt{(\gamma^2+(1-\rho)^2)\ (1+\gamma^2+\rho^2-2\rho\,\mathrm{Cos}[\alpha-\theta])}}\right);$$

```
  (Chop[f /. θ → θ2] - Chop[f /. θ → θ1]) / r]
```

```
LoopField[r_, α_, R_, δ_] := Module[{a, g, l1, l2, t, ω, ξ},
  {a, ω, ξ, g} = {InjectionRadius, MagnetOpening,
     SpiralAngle, MagnetHalfGap} /. FFAGSpiralParameters;
  t = Tan[ξ];
  l1 = Log[R / a];
  l2 = Log[(R + δ) / a];
  PolarConductorField[r, α, R, l1 t, ω + l1 t, g] -
   PolarConductorField[r, α, R + δ, l2 t, ω + l2 t, g]]
```

Out[1039]=
```
{Listable}
```

Out[1040]=
```
{Listable}
```

Out[1041]=
```
{InjectionKineticEnergy → 17, InjectionField → if, InjectionRadius → ir,
 EjectionKineticEnergy → 180, NumberOfCells → 8, PackingFactor → 0.38,
 FieldPower → 4.385, EjectionField → ef, EjectionRadius → 3.4825,
 MagnetOpening → 0.298451, MagnetHalfGap → 0.03, SpiralAngle → 0.86961}
```

Out[1042]=
```
{InitialAngle → 0, MachineCenter → CPoint[0, 0], Segment → False}
```

In[186]:=
```
FFAGSpiralArc[2, 1, 2, π / 6, π / 4]
```

Out[186]=

$$\text{PPoint}\left[2\,\text{Cos}[\#1]\ \&,\ 2\,\text{Sin}[\#1]\ \&,\ \text{PRange}\to\left\{\text{Log}[2],\ \frac{\pi}{6}+\text{Log}[2]\right\}\right]$$

```
LoopField[3, π / 4, 2, 1, π / 6, π / 4, .03, .03]
```

```
0.0256921
```

```
Plot[LoopField[r, π / 4, 2, 1, π / 6, π / 4, .03, .03], {r, 1.5, 2.5}];
```



In[1049]:=
```
ffag = FFAGSpiral[Segment → True];
```

In[1050]:=
```
FFAGSpiralParameters
```

Out[1050]=
```
{NumberOfCells → 8, PackingFactor → 0.38, InjectionKineticEnergy → 17,
 EjectionKineticEnergy → 180, EjectionRadius → 3.4825, FieldPower → 4.385,
 MagnetOpening → 0.298451, SpiralAngle → 0.86961, MagnetHalfGap → 0.03,
 InjectionField → 0.554555, InjectionRadius → 2.77591, EjectionField → 1.499}
```

In[445]:=

```
ffag = FFAGSpiral[8, .38, 17, 180,
   3.4825, 4.385, 17.1 π / 180, 49.825 π / 180, Segment → True];
{{f1, f2}, {a, b}} = {FFAGSpiralFields, FFAGSpiralRadii} /. ffag
Draw[Paint[FFAGSpiralGraph /. ffag, Blue]];
```

Out[446]=

```
{{0.554555, 1.499}, {2.77591, 3.4825}}
```



In[716]:=

```
cd = CurrentDistribution[1];
```

In[907]:=

```
cd = CurrentDistribution[2];
```

In[917]:=

```
cd = CurrentDistribution[2,
   FieldCorrection → ((47.57020841562868` - 59.946083298215186` # +
        28.199263892889444` #^2 - 5.869987636750913` #^3 + 0.456307880820006` #^4) &)];

Currents /. cd
```

Out[920]=

```
{-789075., 555306., -363177., 105426., -110110., -7515.56, -51833.2, -29146.4,
 -37500.4, -31849.2, -32782.6, -30805.6, -30227.9, -29038.6, -28134., -27085.7,
 -26080.6, -25029.2, -23972.5, -22893.5, -21802.5, -20696.8, -19579.2,
 -18449., -17306.6, -16151.1, -14981.7, -13797.4, -12597., -11379., -10142.3,
 -8885.44, -7607.28, -6306.76, -4983.05, -3635.58, -2264.17, -869.026, 549.244,
 1989.61, 3450.92, 4931.91, 6432.45, 7952.68, 9497.53, 11069.8, 12688.6,
 14351.1, 16118.1, 17926.4, 20002.3, 21950.3, 24763.7, 26395.5, 31533.6,
 30054.2, 44556.2, 25527.7, 82861.5, -28267.4, 260540., -491080., 1.09116 × 10^6}
```

In[921]:=

```
Length[%]
```

Out[921]=

```
63
```

In[918]:=
```
fe = FieldError /. cd;
g1 = Plot[fe[r], {r, 2.77, 3.5}, AxesLabel → {"r [m]", "ΔB/B"},
   Epilog → {Blue, PointSize[.02], Point[{2.7759, 0}], Point[{3.4825, 0}]}];
```



In[1052]:=
```
ff = Field3DFunction /. cd;
ff[3, π / 4]
```

Out[1053]=
0.0749732 + 0. i

```
g3 = Plot[y, {x, 2.77, 3.5}, PlotStyle → Magenta];
```

In[908]:=
```
tfe = Table[{r, fe[r]}, {r, 2.77, 3.5, .05}];
y = Fit[Chop[tfe], {1, x, x^2, x^3, x^4}, x]
g2 = ListPlot[tfe, PlotStyle → {PointSize[.02], Blue}];
g3 = Plot[y, {x, 2.77, 3.5}, PlotStyle → Blue];
Show[g1, g2, g3];
```

Out[909]=
$-0.0767372 + 0.10835\,x - 0.0565787\,x^2 + 0.0129754\,x^3 - 0.00110436\,x^4$

In[846]:=
```
tfe = Table[{r, fe[r]}, {r, 2.77, 3.5, .05}];
y = Fit[Chop[tfe], {1, x, x^2, x^3, x^4}, x]
g2 = ListPlot[tfe, PlotStyle → {PointSize[.02], Blue}];
g3 = Plot[y, {x, 2.77, 3.5}, PlotStyle → Magenta];
Show[g1, g2, g3];
```

Out[847]=
$47.5702 - 59.9461\,x + 28.1993\,x^2 - 5.86999\,x^3 + 0.456308\,x^4$




0.002
-0.002
-0.004
-0.006
-0.008
2.8
2.9
3.1
3.2
3.3
3.4
3.5


0.002
-0.002
-0.004
-0.006
-0.008
2.8
2.9
3.1
3.2
3.3
3.4
3.5

Code of the figures

■ Field produced by a conductor

```
c = CPoint[0, 0, 0];
ffag = To3D[FFAGSpiral[1, 1, 3, Pi / 6, Pi / 4]];
a1 = To3D@(FFAGSpiralArc[1, 1, 2, Pi / 6, Pi / 4][[1]]);
a2 = Translate[a1, CPoint[0, 0, .5]];
{m1, m2} = Pointer[{a1, a2}, Log[2]];
{m3, m4} = Pointer[{a1, a2}, Log[2] + π / 6];
s = Segment[{m1, m3}, {m2, m4}];
a3 = To3D@(FFAGSpiralArc[1, 1, 1.5, Pi / 6, Pi / 4][[1]]);
{m5, m51, m6} = Pointer[{a2, a2, a3}, {Log[2] + π / 18, Log[2] + π / 6, .8}];
s1 = Segment[m5, m6];
p = Plane[m5, m51, m6];
d = ELine[m6, p];
m7 = Pointer[d, -1, m6];
s2 = Arrow@Segment[m6, m7];
{m8, m9} = Pointer[ffag[[1, 1]], {0, π/6}];
s3 = Segment[c, {m8, m9}];
s4 = Zoom[s3[[1]], c, 2];
tg = ELine[ffag[[1, 3]], 0];
s5 = Segment @@ Pointer[tg, {0, 1}, m8];
te = Legend[{"C", "M", "A", "B", "I", "ω", "a", "ξ", "g"},
   {c, m5, m6, m7, m51, CPoint[1 / 4, 1 / 20, 0], Pointer[s3[[2]], 1 / 2],
    CPoint[1.2, 1 / 10, 0], Pointer[s[[1]], 1 / 2]}, Offset → {.1, -Pi / 2}];
Draw3D[Black, s2, Red, te, CPoint[0, 0, 0], ffag, Green, a1, s, s3, s4, s5, Blue,
  m5, m6, Arrow[a2, 10], a3, s1, Axes → True, Ticks → None, AxesLabel → {"x", "y", "z"}];
```



■ Full spiral FFAG

```
ffag = FFAGSpiral[8, 1, 2, Pi / 12, Pi / 4, Segment → True];
{a, b, c} = CPoint[{0, 3, 3}, {0, 0, 2}];
an = ArcTan[3, 2];
ax = Segment[a, {b, c}];
{a1, a2} = Pointer[ax, .2];
ci = ECircle[a, {1, 2}];
t1 = Pointer[ci[[1]], π / 2 + {0, π / 12}];
as1 = Segment[a, t1];
t2 = Pointer[ci[[1]], 5 π / 6];
as2 = Segment[a, t2];
ar = Arc[a, a1, a2];
r = Exp[#/Tan[ξ]];
pr = PRange → {t1, t2, 25};
m =
  PPoint[Exp[(# - π / 12)/Tan[Pi / 4]] Cos[#] &, Exp[(# - π / 12)/Tan[Pi / 4]] Sin[#] &, PRange → {0, π / 4, 25}];
x = Intersections[ax[[2]], ci];
x1 = Pointer[m, an];
d = ELine[m, an];
t3 = Pointer[ELine[a, c], 1, x1];
ar1 = Arc[x1, t3, π / 4];
te = Legend[{"r₁", "r₂"}, x, Offset → {3, π / 2}];
te1 = Legend[{"θ", "ω", "(2 π)/n"},
   CPoint[.7 Cos[{an / 2, 13 π / 24, 17 π / 24}], .7 Sin[{an / 2, 13 π / 24, 17 π / 24}]]];
x2 = Pointer[Bisector[ELine[a, c], d][[2]], 1.2, x1];
te2 = Legend["ξ", x2];
Draw[Black, LineOrigin[x1], DrawRange[-2, 2], ci, te, te1, te2, as1, as2,
  Arrow[ax], Arrow[{ar, ar1}], Paint[ffag, Blue], x, Yellow, x1, Green, d];
```

```
ffag = FFAGSpiral[8, 1, 2, Pi / 12, Pi / 4, Segment → True];
{a, b, c} = CPoint[{0, 3, 3}, {0, 0, 2}];
an = ArcTan[3, 2];
ax = Segment[a, {b, c}];
{a1, a2} = Pointer[ax, .2];
ci = ECircle[a, {1, 2}];
t1 = Pointer[ci[[1]], π / 2 + {0, π / 12}];
as1 = Segment[a, t1];
t2 = Pointer[ci[[1]], 5 π / 6];
as2 = Segment[a, t2];
ar = Arc[a, a1, a2];
r = Exp[# / Tan[ξ]];
pr = PRange → {t1, t2, 25};
m =
  PPoint[Exp[(# - π / 12) / Tan[Pi / 4]] Cos[#] &, Exp[(# - π / 12) / Tan[Pi / 4]] Sin[#] &, PRange → {0, π / 4, 25}];
x = Intersections[ax[[2]], ci];
x1 = Pointer[m, an];
d = ELine[m, an];
t3 = Pointer[ELine[a, c], 1, x1];
ar1 = Arc[x1, t3, π / 4];
te = Legend[{"r1", "r2"}, x, Offset → {3, π / 2}];
te1 = Legend[{"θ", "ω", "2 π / n"},
   CPoint[.7 Cos[{an / 2, 13 π / 24, 17 π / 24}], .7 Sin[{an / 2, 13 π / 24, 17 π / 24}]]];
x2 = Pointer[Bisector[ELine[a, c], d][[2]], 1.2, x1];
te2 = Legend["ξ", x2];
Draw[Black, LineOrigin[x1], DrawRange[-2, 2], ci, te, te1, te2, as1, as2,
  Arrow[ax], Arrow[{ar, ar1}], Paint[ffag, Blue], x, Yellow, x1, Green, d];
```